

# ECOLANG

## **Communications Language for Ecological Simulations Network**

*Messages and Communications Protocol for  
EcoSimNet Framework*



*António Pereira*

**Faculty of Engineering of University of Porto**  
*LIACC – Artificial Intelligence and Computer Science Laboratory*

Technical Report TR-LIACC-FEUP-AMCP 01.2

*Language version 1.4 - June 2010*

## Index

Introduction .....	3
1 Message definition .....	3
2 Message types.....	3
2.1 Connection .....	4
2.2 Definitions.....	4
2.3 Actions.....	6
2.4 Perceptions .....	9
2.5 Coordination .....	11
3 Communications Protocol .....	13
4 Software.....	15
4.1 Header Files .....	15
4.2 Library File.....	15
4.3 Dll File .....	15
4.4 Dependencies .....	15
5 Document Versions.....	16
Version 1.4 .....	16
Version 1.3c .....	16
Version 1.3b .....	16
Version 1.3a .....	16
Version 1.3 .....	16
Version 1.2 .....	16
Version 1.1 .....	16
Version 1.0 .....	17
Version 0.4a .....	17
Version 0.4 .....	17
Version 0.3g .....	17
Version 0.3f .....	17
Version 0.3e .....	17
Version 0.3d .....	17
Version 0.3c .....	18
Version 0.3b .....	18
Version 0.3 .....	18
Version 0.2 .....	18
Version 0.1 .....	18
6 References.....	18

## Introduction

This document (“ECOLANG\_v\_1\_4\_Eng.docx”) describes the communication language used in the EcoSimNet framework, a multi-agent systems environment for ecological simulations, based on EcoDynamo simulator (Pereira and Duarte, 2005), several intelligent agents and visualisation applications, and extends the initial definition of the language (Pereira et al., 2005).

The agents’ actions and perceptions are translated into messages exchanged between the simulator application and the agents.

The concepts’ definitions used follow the BNF notation (Backus et al., 1960) and they’re inspired in the Coach Unilang language (Reis and Lau, 2002).

ECOLANG notation is an extension to the original BNF formalism adding the following meta-symbols:

{ } used for repetitive items (one or more times);

[ ] encloses types of values;

Terminal symbols use bold face letters.

## 1 Message definition

The base syntax of each message is as follows:

```
<MESSAGE> ::= message (<ID> <SENDER> <RECEIVER> <MSG_CONTENT>)
<ID> ::= [integer]
<SENDER> ::= [string]
<RECEIVER> ::= [string]
```

<ID> is the message identifier sent by the initiator – it is a sequential integer number controlled by each sender (initial value is 1).

<SENDER> is the name of the message initiator application (source).

<RECEIVER> is the name of the message destination application.

<MSG\_CONTENT> is the content of the message.

Each message will be represented by a numeric reference to facilitate its identification.

## 2 Message types

Message exchanged by the applications may belong to five types: connection, definitions, actions, perceptions, and coordination.

The connection messages establish the communication sessions between the agents / applications and specify the computer where agent belongs and the port number in which the agent accepts connections and listens the messages.

Though there could be actions and perceptions dedicated to some type of the agents belonging to the system, there are no restrictions to the messages that applications can use.

```
<MSG_CONTENT> ::= <CONNECTION_MSG> | <DEFINITION_MSG> | <ACTION_MSG> |
<PERCEPTION_MSG> | <COORDINATION_MSG>
```

## 2.1 Connection

Connection messages define the start and the finish of the communications sessions between applications. In this group there are also messages to ask the agents known by the other partner of the session. This allows the establishment of links between multiple applications, facilitating the expansion of the communications and knowledge network.

```
<CONNECTION_MSG> ::= <CONNECT> | <DISCONNECT> | <ACCEPT> | <ASK_AGENTS> |
<KNOWN_AGENTS>
```

The session begins with the *connect* message and finishes with the *disconnect* message. The *accept* message answers any of those messages:

```
<CONNECT> ::= connect <HOST_NAME> <HOST_ADDR> <SERVER_PORT> [1.1]
<DISCONNECT> ::= disconnect [1.2]
<ACCEPT> ::= accept (<ACTION_ID> <ACTION_RESULT>) [1.3]
```

```
<HOST_NAME> ::= [string]
<HOST_ADDR> ::= [string]
<SERVER_PORT> ::= [integer]
<ACTION_ID> ::= <ID>
<ACTION_RESULT> ::= ok | failed
```

To ask the other partner for the known agents:

```
<ASK_AGENTS> ::= agents [1.4]
<KNOWN_AGENTS> ::= known_agents (<ACTION_ID> {<AGENT>}) [1.5]
```

```
<AGENT> ::= (<AGENT_NAME> <HOST_NAME> <HOST_ADDR> <SERVER_PORT> <CONNECTED>)
<AGENT_NAME> ::= [string]
<CONNECTED> ::= connected | disconnected
```

<HOST\_NAME> and <HOST\_ADDR> identify the computer name and IP address where the application runs, <SERVER\_PORT> identifies the port number where the application expects connections.

<ACTION\_ID> of the answer messages must be the <ID> of the corresponding connection message.

## 2.2 Definitions

From version 1.3 of the protocol, these messages include the definition of regions and information about the type of model loaded in the simulator, its dimensions, its morphology and species of shellfish and molluscs prevailing in it.

```
<DEFINITION_MSG> ::= <REGIONS_MSG> | <MODEL_DEFINITIONS>
```

Each region is referred to by name, will be of one type and will cover a certain area. It may also be defined at the expense of other regions already defined.

```

<REGIONS_MSG> ::= <DEFINE_ACTION> | <DELETE_ACTION> | <GET_REGIONS> | <GET_REGION>
  | <DEFINE_RESULT> | <DELETE_RESULT> | <GET_REGIONS_RESULT> |
  <GET_REGION_RESULT>

<DEFINE_ACTION> ::= define (<REG_NAME> <REGION>) [2.1]
<DELETE_ACTION> ::= delete {<REG_NAME>} [2.2]

<REG_NAME> ::= [string]
<REGION> ::= <REGION_TYPE> <REGION_AREA> | {<REG_NAME>}

```

Each region is of the type land or water and, in the later, is characterized by its quality as well as the type and quality of its sediment.

```

<REGION_TYPE> ::= <LAND_REGION> | <WATER_REGION>
<LAND_REGION> ::= land
<WATER_REGION> ::= <WATER_CARACT> <SEDIMENT_CARACT>
<WATER_CARACT> ::= <SUB_INTERTIDAL> <WATER_QUALITY>
<SUB_INTERTIDAL> ::= subtidal | intertidal
<WATER_QUALITY> ::= <QUAL_SCALE>
<QUAL_SCALE> ::= excellent | good | poor
<SEDIMENT_CARACT> ::= (<SEDIMENT_TYPE> <SEDIMENT_QUALITY>)
<SEDIMENT_TYPE> ::= sandy | sand_muddy | muddy
<SEDIMENT_QUALITY> ::= <QUAL_SCALE>

```

The region area is a set of one or more simple regions, each one defined by one point or a simple polygon (rectangle, square, circle or circle arc).

```

<REGION_AREA> ::= {<SIMPLE_REGION>}
<SIMPLE_REGION> ::= <POINT> | (rect <POINT> <POINT>) | (square <POINT> <POINT>
  <POINT> <POINT>) | (circle <POINT> [real]) | (arc <POINT> [real] [real] [real]
  [real])
<POINT> ::= (point [integer] [integer])

```

Each of the previous messages (**define** and **delete**) should be answered by the application that receive the message:

```

<DEFINE_RESULT> ::= define_result (<ACTION_ID> <ACTION_RESULT>) [2.3]
<DELETE_RESULT> ::= delete_result (<ACTION_ID> <ACTION_RESULT>) [2.4]

```

At any time it is possible to know which are the defined regions and their characteristics:

```

<GET_REGIONS> ::= get_region_names [2.5]
<GET_REGION> ::= get_region <REG_NAME> [2.6]

```

The previous messages should obtain as answers, respectively

```

<GET_REGIONS_RESULT> ::= region_names (<ACTION_ID> {<REG_NAME>}) [2.7]
<GET_REGION_RESULT> ::= region (<ACTION_ID> <REG_NAME> <REGION>) [2.8]

```

There are also messages to get information about the model loaded in the simulator – they can obtain the size, type, morphology and species of molluscs and shellfish prevailing in it:

```
<MODEL_DEFINITIONS> ::= <GET_DIMENSIONS> | <GET_MORPHOLOGY> | <GET_SPECIES> |
    <DIMS_RESULT> | <MORPHOLOGY_RESULTS> | <SPECIES_RESULTS>

<GET_DIMENSIONS> ::= model_dimensions [2.9]
<GET_MORPHOLOGY> ::= model_morphology [2.10]
<GET_SPECIES> ::= model_species [2.11]
```

The previous messages should obtain as answers, respectively:

```
<DIMS_RESULT> ::= dimensions (<ACTION_ID> <LINES> <COLUMNS> <LAYERS> <MOD_TYPE>) [2.12]
<MORPHOLOGY_RESULTS> ::= <MORPHOLOGY_RESULT> | <MORPHOLOGY_END>
<MORPHOLOGY_RESULT> ::= morphology (<ACTION_ID> > {(<CELL> [real])}) [2.13]
<MORPHOLOGY_END> ::= morphology_end [2.14]
<SPECIES_RESULTS> ::= <SPECIES_RESULT> | <SPECIES_END>
<SPECIES_RESULT> ::= benthic_species (<ACTION_ID> {(<SPECIES_NAME> <BOXES>)}) [2.15]
<SPECIES_END> ::= benthic_species_end [2.16]

<LINES> ::= [integer]
<COLUMNS> ::= [integer]
<LAYERS> ::= [integer]
<MOD_TYPE> ::= 0D | 1DH | 1DV | 2DH | 2DV | 3D
<CELL> ::= [integer]
<SPECIES_NAME> ::= ([string])
<BOXES> → defined in message [3.13] (<GET_VAR_VALUE>)
```

## 2.3 Actions

The actions' messages are closely linked to each type of agent involved in the system.

An agent / application that has an interest in the production of shellfish has actions of deposit, inspect and collect species of molluscs.

```
<ACTION_MSG> ::= <SEED_ACTION> | <INSPECT_ACTION> | <HARVEST_ACTION> | <ACTION_SIM>
```

To deposit (seed), the agent indicates the region, the time, the characteristics of the species of molluscs to deposit and the total weight seeded. The two real values indicated in the message may have different meanings, depending on molluscs in question. By example, for the oysters and scallops, the first value indicates the individual weight of the shell and the second indicates the individual weight of meat; for clams, the first value indicates the individual dry weight, and the second indicates the individual weight.

To inspect, the agent indicates the region and the time of the inspection.

To collect (harvest), it must indicate, beyond the region, the characteristics of shellfish to collect and time of collection.

<SEED\_ACTION> ::= **seed** (<REG\_NAME> <TIME> <BIVALVE\_S> <DENSITY>) [3.1]

<INSPECT\_ACTION> ::= **inspect** (<REG\_NAME> <TIME>) [3.2]

<HARVEST\_ACTION> ::= **harvest** (<REG\_NAME> <TIME> <BIVALVE>) [3.3]

<BIVALVE\_S> ::= <BTYPE> ([real] [real])

<BIVALVE> ::= <BTYPE> <SHELL\_LENGTH>

<BTYPE> ::= **scallop** | **kelp** | **oyster** | **mussel** | **clam**

<SHELL\_LENGTH> ::= (**length** [real])

<DENSITY> ::= (**density** [real])

Reference to the action moment can be as quickly as possible (*now*) or one value that indicates the number of seconds from 1970 January, 1 00:00:00.

<TIME> ::= **now** | [integer]

Any agent / application can act over the simulator choosing the model it wants to simulate, controlling the parameterization of the model - gathering / changing parameters of the simulated classes and collecting / recording the results of the simulation. Messages can be divided into four different types:

<ACTION\_SIM> ::= <MODEL\_ACTION> | <EXEC\_ACTION> | <SPECS\_ACTION> | <REG\_ACTION>

- Actions to choose the model to simulate - open or close model, survey the model in simulation

<MODEL\_ACTION> ::= <OPEN\_MODEL> | <CLOSE\_MODEL> | <GET\_MODEL> | <SAVE\_CONF>

<OPEN\_MODEL> ::= **open\_model** <MODEL\_NAME> [3.4]

<CLOSE\_MODEL> ::= **close\_model** [3.5]

<GET\_MODEL> ::= **model\_name** [3.6]

<SAVE\_CONF> ::= **save\_configuration** [3.7]

<MODEL\_NAME> ::= [string]

- Actions over the simulation execution - monitor and influence the simulations - initialising, running, stopping, restarting and completing the simulations:

<EXEC\_ACTION> ::= **initialise** | **run** | **stop** | **pause** | <STEP\_CMD> [3.8]

<STEP\_CMD> ::= **step** [integer] [3.9]

- Actions over the model in simulation – to choose / survey classes, choose / inquire initial values for variables and parameters, choose / survey frequency and range of simulation, choose / survey simulation sub-domain:

<SPECS\_ACTION> ::= <SP\_CLASSES> | <SP\_VARS> | <SP\_PARMS> | <SP\_TIME> | <SUB\_DOMAIN>

<SP\_CLASSES> ::= <GET\_CLASSES> | <SELECT\_CLASSES>

<GET\_CLASSES> ::= **get\_available\_classes** | **get\_selected\_classes** [3.10]

<SELECT\_CLASSES> ::= **select\_classes** {<CLASS\_NAME>} [3.11]

<SP\_VARS> ::= <GET\_CLASS\_VARS> | <GET\_VAR\_VALUE> | <SET\_VAR\_VALUE>

<GET\_CLASS\_VARS> ::= **get\_variables** <CLASS\_NAME> [3.12]

<GET\_VAR\_VALUE> ::= **get\_variable\_value** <CLASS\_NAME> <VAR\_NAME> <BOXES> [3.13]

<SET\_VAR\_VALUE> ::= **set\_variable\_value** <CLASS\_NAME> {(<VAR\_NAME> <BOXES> [real])} [3.14]

<SP\_PARMS> ::= <GET\_PARMS> | <SET\_PARMS>

<GET\_PARMS> ::= **get\_parameters** <CLASS\_NAME> [3.15]

<SET\_PARMS> ::= **set\_parameters** <CLASS\_NAME> {(<PARAM\_NAME> [real])} [3.16]

<SP\_TIME> ::= <GET\_TIME> | <SET\_TIME>

<GET\_TIME> ::= **get\_time\_spec** [3.17]

<SET\_TIME> ::= **set\_time\_spec** <STEP> <START\_TIME> <FINISH\_TIME> [3.18]

<SUB\_DOMAIN> ::= **subdomain** <DOMAIN> [3.19]

<CLASS\_NAME> ::= ([string])

<VAR\_NAME> ::= ([string])

<BOXES> ::= (<SUB\_DOMAIN>) | ({<CELL>})

<PARAM\_NAME> ::= ([string])

<STEP> ::= [integer]

<START\_TIME> ::= [integer]

<FINISH\_TIME> ::= [integer]

<DOMAIN> ::= **all** | ({<REG\_NAME>})

- Actions to record the results - choose variables to register, frequency, range and type of recording, choose sub-domain to register or activate the monitoring mode (trace):

<REG\_ACTION> ::= <REG\_FILE> | <REG\_VARS> | <REG\_LOG> | <REG\_TIME> | <REG\_TRACE>

<REG\_FILE> ::= **output\_file** <FILE\_NAME> [3.20]

<REG\_VARS> ::= <GET\_VARS> | <SELECT\_VARS> | <UNSELECT\_VARS>

<GET\_VARS> ::= **get\_available\_variables** [3.21]

<SELECT\_VARS> ::= **select\_variables** <OUTPUT\_TYPE> ({<VAR\_NAME>}) <BOXES> [3.22]

<UNSELECT\_VARS> ::= **unselect\_variables** <OUTPUT\_TYPE> {<VAR\_NAME>} [3.23]

<REG\_LOG> ::= **log** <LOG\_TYPE> ({<LOG\_STEP>}) [3.24]

<REG\_TIME> ::= <GET\_REG\_TIME> | <SET\_REG\_TIME>

<GET\_REG\_TIME> ::= **get\_output\_time** [3.25]

<SET\_REG\_TIME> ::= **set\_output\_time** <STEP> <START\_TIME> <FINISH\_TIME> [3.26]

<REG\_TRACE> ::= **trace** [3.27]

<FILE\_NAME> ::= ([string])



```

<OUTPUT_TYPE> ::= file | graph | table | remote
<LOG_TYPE> ::= xml | xls | txt | remote
<LOG_STEP> ::= [integer]

```

## 2.4 Perceptions

The perceptions' messages are closely related with the type of agent involved in the system and also to the actions performed by each over the simulator.

An agent with an interest in the production of shellfish, seed molluscs and its perceptions are the result of the actions taken.

The response to the seed action of the agent may be positive or negative (in the case such action is denied). In response to the inspection action the agent receives a message with the bivalve's characteristics in the region. The resulting harvest is negative or positive, and in this case, it is indicated the total weight harvested.

```

<PERCEPTION_MSG> ::= <SEED_RESULT> | <INSPECT_RESULT> | <HARVEST_RESULT> |
<PERCEPTION_SIM>

<SEED_RESULT> ::= seed_result (<ACTION_ID> <ACTION_RESULT>) [4.1]
<INSPECT_RESULT> ::= inspect_result (<ACTION_ID> {<BIVALVE>}) [4.2]
<HARVEST_RESULT> ::= harvest_result (<ACTION_ID> <ACTION_RESULT> <WEIGHT>) [4.3]

<WEIGHT> ::= [real]

```

The <ACTION\_ID> of the perception message identifies the <ID> of the corresponding action message.

The agents' / applications' perceptions are both messages with the result of the actions initiated by the agent / application or messages spontaneously sent by the simulator. They may be from five types:

```

<PERCEPTION_SIM> ::= <MODEL_RESULT> | <EXEC_RESULT> | <SPECS_RESULT> | <REG_RESULT>
| <EVENT_MSG>

```

- Answers to the actions over the model:

```

<MODEL_RESULT> ::= <OPEN_RESULT> | <CLOSE_RESULT> | <GET_RESULT> | <SAVE_RESULT>
<OPEN_RESULT> ::= open_result (<ACTION_ID> <ACTION_RESULT>) [4.4]
<CLOSE_RESULT> ::= close_result (<ACTION_ID> <ACTION_RESULT>) [4.5]
<GET_RESULT> ::= model (<ACTION_ID> <MODEL_NAME>) [4.6]
<SAVE_RESULT> ::= save_result (<ACTION_ID> <ACTION_RESULT>) [4.7]

```

- Answers to the actions over the simulation execution:

```

<EXEC_RESULT> ::= exec_result (<ACTION_ID> <ACTION_RESULT>) [4.8]

```

- Answers to the actions over the model configuration:

```

<SPECS_RESULT> ::= <CLASSES_RESULT> | <VARS_RESULT> | <PARMS_RESULT> |
  <TIME_RESULT> | <SUB_DOMAIN_RESULT>

<CLASSES_RESULT> ::= <CLASSES_AVAILABLE> | <CLASSES_SELECTED>
<CLASSES_AVAILABLE> ::= classes_available (<ACTION_ID> {<CLASS_NAME>}) [4.9]
<CLASSES_SELECTED> ::= classes_selected (<ACTION_ID> {<CLASS_NAME>}) [4.10]

<VARS_RESULT> ::= <CLASS_VARS> | <VAR_VALUE> | <VAR_SET>
<CLASS_VARS> ::= variables (<ACTION_ID> {<VAR_NAME>}) [4.11]
<VAR_VALUE> ::= variable_value (<ACTION_ID> {(<CELL> [real])}) [4.12]
<VAR_SET> ::= variable_set_result (<ACTION_ID> <ACTION_RESULT>) [4.13]

<PARMS_RESULT> ::= <CLASS_PARMS> | <PARMS_SET>
<CLASS_PARMS> ::= parameters_values (<ACTION_ID> {(<PARAM_NAME> [real])}) [4.14]
<PARMS_SET> ::= parameters_set_result (<ACTION_ID> <ACTION_RESULT>) [4.15]

<TIME_RESULT> ::= time_spec (<ACTION_ID> <STEP> <START_TIME> <FINISH_TIME>) [4.16]

<SUB_DOMAIN_RESULT> ::= subdomain_result (<ACTION_ID> <ACTION_RESULT>) [4.17]

```

- Answers to the actions over the register of the results:

```

<REG_RESULT> ::= <FILE_RESULT> | <REG_VARS_RESULT> | <LOG_RESULT> |
  <REG_TIME_RESULT> | <TRACE_RESULT>

<FILE_RESULT> ::= output_file_result (<ACTION_ID> <ACTION_RESULT>) [4.18]

<REG_VARS_RESULT> ::= <GET_VARS_RESULT> | <SELECT_VARS_RESULT> |
  <UNSELECT_VARS_RESULT>
<GET_VARS_RESULT> ::= variables_available (<ACTION_ID> {<VAR_NAME>}) [4.19]
<SELECT_VARS_RESULT> ::= select_variables_result (<ACTION_ID> <ACTION_RESULT>) [4.20]
<UNSELECT_VARS_RESULT> ::= unselect_variables_result (<ACTION_ID> <ACTION_RESULT>) [4.21]

<LOG_RESULT> ::= log_result (<ACTION_ID> <ACTION_RESULT>) [4.22]
<REG_TIME_RESULT> ::= output_time (<ACTION_ID> <STEP> <START_TIME> <FINISH_TIME>) [4.23]

<TRACE_RESULT> ::= trace_result (<ACTION_ID> <TRACE_STATUS>) [4.24]
<TRACE_STATUS> ::= on | off

```

- Spontaneous messages sent by the simulator:

```

<EVENT_MSG> ::= <REG_MSG> | <LOG_MSG> | <END_MSG>

<REG_MSG> ::= register (<REG_INDEX> <REG_TIME> <VAR_NAME> {(<CELL> [real])}) [4.25]
<LOG_MSG> ::= logger (<STEP> {(<CLASS_NAME> <FUNC_TYPE> <DATA_CLASS> <VAR_NAME>
  <CELL> [real])}) [4.26]
<END_MSG> ::= end_simulation | end_step | running | stopped | paused [4.27]

```

```

<REG_INDEX> ::= [integer]
<REG_TIME> ::= [integer]
<FUNC_TYPE> ::= Inquiry | Update
<DATA_CLASS> ::= <CLASS_NAME>

```

## 2.5 Coordination

The coordination messages are used between agents and can be grouped by type. The first type includes messages for optimization and for parallel simulated annealing (PSA) coordination.

The PSA algorithm defines one coordinator agent and several workers, each one responsible for a group of simulators. The messages are divided into four groups: to define agents' roles, to configure the simulations and the desired optimization, and to exchange agents' optimization data.

```

<COORDINATION_MSG> ::= <FUNCTION_MSG> | <CONFIGURATION_MSG> | <PROCESS_MSG> |
<OPTIMIZATION_MSG>

```

Each <COORDINATION\_MSG> must be answered by one <EXEC\_RESULT> [4.8] perception.

- Actions to configure the role of each agent:

```

<FUNCTION_MSG> ::= <COORDINATOR_MSG> | <WORKER_MSG> | <SIMULATORS_MSG>

```

```

<COORDINATOR_MSG> ::= psa_coordinator [5.1]

```

```

<WORKER_MSG> ::= psa_worker <COORDINATOR_NAME> [5.2]

```

```

<SIMULATORS_MSG> ::= psa_simulator ({<SIMULATOR_NAME>}) [5.3]

```

```

<COORDINATOR_NAME> ::= [string]

```

```

<SIMULATOR_NAME> ::= [string]

```

All the <FUNCTION\_MSG> messages are sent from the configuration agent to the agents of the PSA process. The message <COORDINATOR\_MSG> is sent to the coordinator agent. The message <WORKER\_MSG> is sent to the coordinator and worker agents. The message <SIMULATORS\_MSG> is sent to each one of the workers.

- Actions to configure the simulations and the optimization process:

```

<CONFIGURATION_MSG> ::= <TEMPERATURE_MSG> | <ITERATIONS_MSG> | <WORKERS_MSG> |
<SYNCH_POINTS_MSG>

```

```

<TEMPERATURE_MSG> ::= psa_temperature (<INITIAL_T> <FINAL_T> <DRATE_T>) [5.4]

```

```

<ITERATIONS_MSG> ::= psa_iterations [integer] [5.5]

```

```

<WORKERS_MSG> ::= psa_workers [integer] [5.6]

```

```

<SYNCH_POINTS_MSG> ::= psa_synchronization_points [integer] [5.7]

```

```

<INITIAL_T> ::= [real]

```

```

<FINAL_T> ::= [real]

```

```

<DRATE_T> ::= [real]

```

All the `<CONFIGURATION_MSG>` messages are sent from the configuration agent to the agents of the PSA process. The messages `<TEMPERATURE_MSG>` and the `<ITERATIONS_MSG>` are sent to the coordinator and worker agents. The messages `<WORKERS_MSG>` and the `<SYNCH_POINTS_MSG>` are sent to the coordinator agent.

- Actions to start optimization and to exchange agents' results:

```
<PROCESS_MSG> ::= <PUBLISH_MSG> | <RESULTS_MSG> | <RUN_MSG>
```

```
<PUBLISH_MSG> ::= <PUBLISH_BEST_RESULT> | <PUBLISH_BEST_SOLUTION>
```

```
<PUBLISH_BEST_RESULT> ::= psa_publish_best_result [real] [5.8]
```

```
<PUBLISH_BEST_SOLUTION> ::= psa_publish_best_solution ({<SOLUTION>}) [5.9]
```

```
<RESULTS_MSG> ::= <BEST_RESULT> | <BEST_SOLUTION>
```

```
<BEST_RESULT> ::= psa_best_result [real] [5.10]
```

```
<BEST_SOLUTION> ::= psa_best_solution ({<SOLUTION>}) [5.11]
```

```
<RUN_MSG> ::= psa_run [5.12]
```

```
<SOLUTION> ::= (<CELL> [real])
```

```
<FINAL_T> ::= [real]
```

```
<DRATE_T> ::= [real]
```

```
<CELL> → defined in message [2.13] (<MORPHOLOGY_RESULT>)
```

The message `<RUN_MSG>` is initially sent from the configuration agent to the coordinator. The `<RESULTS_MSG>` messages are used by the coordinator and the workers agents to exchange the best results, and the messages `<PUBLISH_MSG>` are sent from the coordinator agent to all the others participants, including the configuration agent, in the end of the optimization process.

- Actions to configure the optimization:

```
<OPTIMIZATION_MSG> ::= <SIMULATION_MSG> | <SPECIES_CONFIGURATION_MSG>
```

```
<SIMULATION_MSG> ::= simulation_steps [integer] [5.13]
```

```
<SPECIES_CONFIGURATION_MSG> ::= species_optimization ({<BIVALVE_S> <DENSITY>  
<NR_CELLS> <ECONOMICAL_WEIGHT> <REG_NAME> <OPT_TYPE>}) [5.14]
```

```
<BIVALVE_S> → defined in message [3.1] (<SEED_ACTION>)
```

```
<DENSITY> → defined in message [3.1] (<SEED_ACTION>)
```

```
<NR_CELLS> ::= [integer]
```

```
<ECONOMICAL_WEIGHT> ::= [real]
```

```
<REG_NAME> → defined in message [2.1] (<DEFINE_ACTION>)
```

```
<OPT_TYPE> ::= areas | density
```

All the `<OPTIMIZATION_MSG>` messages are sent from the configuration agent to the coordinator and worker agents of the PSA process.

### 3 Communications Protocol

The communication between all the EcoSimNet actors present in the simulation system is usually of the type handshake - a message-type action expects to receive an answer from the destination application; that response comes in the form of a perception type message.

Only the spontaneous messages and the logging results sent by the simulator don't require feedback.

The first message of each agent for the simulator must be connected (connect). The reception of a positive acceptance message (to accept ok result) indicates that the agent was registered in the simulator as an agent interested in obtaining results from the simulations. When the agent leaves the system it must send the message to disconnect from the simulator.

The simulator, before leaving the system, must send to all registered active agents the disconnect message.

The communications' protocol establishes the answers expected for each action, according to the following table:

<u>[msg]</u>	<u>Message</u>	<u>Expected answer</u>	<u>[msg]</u>
[1.1]	connect	accept	[1.3]
[1.2]	disconnect	accept	[1.3]
[1.4]	agents	known_agents	[1.5]
[2.1]	define	define_result	[2.3]
[2.2]	delete	delete_result	[2.4]
[2.5]	get_region_names	region_names	[2.7]
[2.6]	get_region	region	[2.8]
[2.9]	model_dimensions	dimensions	[2.12]
[2.10]	model_morphology	morphology <sup>1</sup>	[2.13]
[2.10]	model_morphology	morphology_end <sup>2</sup>	[2.14]
[2.11]	model_species	benthic_species <sup>3</sup>	[2.15]
[2.11]	model_species	benthic_species_end <sup>4</sup>	[2.16]
[3.1]	seed	seed_result	[4.1]
[3.2]	inspect	inspect_result	[4.2]
[3.3]	harvest	harvest_result	[4.3]
[3.4]	open_model	open_result	[4.4]
[3.5]	close_model	close_result	[4.5]
[3.6]	model_name	model	[4.6]
[3.7]	save_configuration	save_result	[4.7]
[3.8]	initialise	exec_result	[4.8]

<sup>1</sup> This is the answer while there were messages to send from morphology: morphology of each message has, at most, 750 elements.

<sup>2</sup> This is the answer indicating end of morphology messages.

<sup>3</sup> This is the answer while there were messages to send from benthic species: each benthic species message has, at most, 150 elements.

<sup>4</sup> This is the answer indicating end of benthic species messages.

<u>[msg]</u>	<u>Message</u>	<u>Expected answer</u>	<u>[msg]</u>
[3.8]	run	exec_result	[4.8]
[3.8]	stop	exec_result	[4.8]
[3.8]	pause	exec_result	[4.8]
[3.9]	step	exec_result	[4.8]
[3.10]	get_available_classes	classes_available	[4.9]
[3.10]	get_selected_classes	classes_selected	[4.10]
[3.11]	select_classes	classes_selected	[4.10]
[3.12]	get_variables	variables	[4.11]
[3.13]	get_variable_value	variable_value	[4.12]
[3.14]	set_variable_value	variable_set_result	[4.13]
[3.15]	get_parameters	parameters_values	[4.14]
[3.16]	set_parameters	parameters_set_result	[4.15]
[3.17]	get_time_spec	time_spec	[4.16]
[3.18]	set_time_spec	time_spec	[4.16]
[3.19]	subdomain	subdomain_result	[4.17]
[3.20]	output_file	output_file_result	[4.18]
[3.21]	get_available_variables	variables_available	[4.19]
[3.22]	select_variables	select_variables_result	[4.20]
[3.23]	unselect_variables	unselect_variables_result	[4.21]
[3.24]	log	log_result	[4.22]
[3.25]	get_output_time	output_time	[4.23]
[3.26]	set_output_time	output_time	[4.23]
[3.27]	trace	trace_result	[4.24]
[4.25]	register	--	
[4.26]	logger	--	
[4.27]	end_simulation	--	
[4.27]	end_step	--	
[4.27]	running	--	
[4.27]	stopped	--	
[4.27]	paused	--	
[5.1]	psa_coordinator	exec_result	[4.8]
[5.2]	psa_worker	exec_result	[4.8]
[5.3]	psa_simulator	exec_result	[4.8]
[5.4]	psa_temperature	exec_result	[4.8]
[5.5]	psa_iterations	exec_result	[4.8]
[5.6]	psa_workers	exec_result	[4.8]

<u>[msg]</u>	<u>Message</u>	<u>Expected answer</u>	<u>[msg]</u>
[5.7]	psa_synchronization_points	exec_result	[4.8]
[5.8]	psa_publish_best_result	exec_result	[4.8]
[5.9]	psa_publish_best_solution	exec_result	[4.8]
[5.10]	psa_best_result	exec_result	[4.8]
[5.11]	psa_best_solution	exec_result	[4.8]
[5.12]	psa_run	exec_result	[4.8]
[5.13]	simulation_steps	exec_result	[4.8]
[5.14]	species_optimization	exec_result	[4.8]

## 4 Software

The code with the implementation of the communications protocol is provided in one C++ DLL (Dynamic Link Library) that can be imported by any application. The DLL is part of the package distribution of the **EcoDynamo** (simulation application).

### 4.1 Header Files

The header files contain the definition of the `EcoDynProtocol` class, the message symbols and the data structures used.

Folder: DLLs/ECDProtocol

Files: `EcoDynProtocol.h`, `ECDPMessages.h`, `ECDPAgents.h`, `AgentsTable.h` e `Region.h`.

Note: the file `EcoDynProtocol.h` includes the other ones.

### 4.2 Library File

Folder: DLLs/ECDProtocol/Lib

File: `ECDP.lib`

### 4.3 DLL File

Folder: DLLs/ECDProtocol/Lib

File: `ECDP.dll`

### 4.4 Dependencies

The library file with the protocol code uses the `Queue` and `Parser` classes that belongs to the `Utilities` library.

Folder: DLLs/Utilities/Lib

File: `Utilities.dll`

Folder: DLLs/Utilities

Header Files: `Queue.h`, `Parser.h`

## 5 Document Versions

### **Version 1.4**

Date: 30.June.2010

Added all the coordination messages.

### **Version 1.3c**

Date: 14.May.2008

Messages added:

- **agents**: ask the known agents;
- **known\_agents**: list with the known agents.

### **Version 1.3b**

Date: 03.July.2007

Messages added:

- **save\_configuration**: saves the actual configuration of the model in the simulator;
- **save\_result**: confirms the save operation in the simulator.

### **Version 1.3a**

Date: 15.May.2007

Messages added:

- **morphology\_end**: end of morphology messages;
- **benthic\_species\_end**: end of benthic species message.

### **Version 1.3**

Date: 30.April.2007

Messages **get\_variable\_value** and **variable\_value** changed.

Messages added:

- **get\_region\_names** and **region\_names**: regions defined in the application;
- **get\_region** and **region**: region characteristics;
- **model\_dimensions** and **dimensions**: dimensions of the simulation model;
- **model\_morphology** and **morphology**: morphology of the simulation model
- **model\_species** and **benthic\_species**: benthic species of the simulation model.

### **Version 1.2**

Date: 15.March.2007

Messages added: **register** and **logger**.

### **Version 1.1**

Date: 26.February.2007

Messages added:

- **running**: simulation running;



- **stopped**: simulation aborted / stopped;
- **paused**: pause in simulation.

### **Version 1.0**

Date: 29.January.2007

Messages added:

- **delete**: remove one defined region;
- **define\_result** and **delete\_result**: result of define and delete actions;
- **unselect\_variables**: unselect variables from register values;
- **unselect\_variables\_result**: result of the previous action.

### **Version 0.4a**

Date: 27.June.2006

Messages changed:

- **seed**: type <BIVALVE> replaced by a new type <BIVALVE\_S> that includes two real values (bivalve type can change their meaning);
- **step**: one more field included to indicate how many steps to run.

Included two new spontaneous messages sent by the simulator indicating end of step or end of simulation (**end\_step** and **end\_simulation**).

### **Version 0.4**

Date: 16.May.2006

Included two new messages to enable the connection from several agents and applications in the same system or computer.

New group included:

- group <CONNECTION\_MSG> with messages: **connect**, **disconnect**, **accept**.

### **Version 0.3g**

Date: 23.January.2006

Type **clam** added to the variable <BTYPE>.

### **Version 0.3f**

Date: 21.December.2005

Changed the message **select\_variables**: type **remote** included in <OUTPUT\_TYPE> and <SUB\_DOMAIN> replaced by <BOXES>.

Changed the message **log**: type **remote** included in <LOG\_TYPE>.

### **Version 0.3e**

Date: 21.October.2005

Changed the definition of the messages **seed** and **harvest**.

### **Version 0.3d**

Date: 03.October.2005

Changed the definition of **point**. Fields are now integers.

### **Version 0.3c**

Date: 27.June.2005

Changed the messages:

- **variable\_value**: removed the fields <VAR\_NAME> and <CELL>;
- **trace\_result**: included field <ACTION\_ID>;
- corrected field <REG\_TIME>.

Chapters renumbered and protocol table included.

### **Version 0.3b**

Date: 05.May.2005

Changed the message **parameters\_value** to **parameters\_values**. Included **mussel** in the <BTYPE> definition.

- **variable\_value**: removed the fields <VAR\_NAME> and <CELL>;
- **trace\_result**: included field <ACTION\_ID>;
- corrected field <REG\_TIME>.

Chapters renumbered and protocol table included.

### **Version 0.3**

Date: 11.February.2005

Included the groups <ACTION\_SIM> and <SIMULATION\_SIM>.

### **Version 0.2**

Date: 27.May.2004

Document reformulation.

### **Version 0.1**

Date: 21.May.2004

Initial version.

## **6 References**

- Backus, J. W., Bauer, F. L., Green, J., Katz, C., McCarthy, J., Perlis, A. J., Rutishauser, H., Samelson, K., Vauquois, B., Wegstein, J. H., Wijngaarden, A. v., and Woodger, M. 1960. "Report on the algorithmic language ALGOL 60". *Commun. ACM*, 3 (5): 299-314.
- Pereira, A., and Duarte, P. 2005. *EcoDynamo - Ecological Dynamics Model Application*. University Fernando Pessoa, Porto (January).
- Pereira, A., Duarte, P., and Reis, L. P. 2005. "ECOLANG - A communication language for simulations of complex ecological systems". In *Proceedings of the 19th European Conference on Modelling and Simulation (ECMS 2005)*, (Riga, Latvia). 493-500.
- Reis, L. P., and Lau, N. 2002. "COACH UNILANG - A Standard Language for Coaching a (Robo)Soccer Team". In *Proceedings of the Robocup 2001 - Robot Soccer World Cup*, (Seattle, WA, USA). Springer-Verlag, 183-192.